Errata

Document Title: Impedance Characterization of Resonators Using the 4194A Impedance / Gain-Phase Analyzer (AN 339-1)

Part Number: 5950-2882

Revision Date: October 1985

HP References in this Application Note

This application note may contain references to HP or Hewlett-Packard. Please note that Hewlett-Packard's former test and measurement, semiconductor products and chemical analysis businesses are now part of Agilent Technologies. We have made no changes to this application note copy. The HP XXXX referred to in this document is now the Agilent XXXX. For example, model number HP8648A is now model number Agilent 8648A.

About this Application Note

We've added this application note to the Agilent website in an effort to help you support your product. This manual provides the best information we could find. It may be incomplete or contain dated information, and the scan quality may not be ideal. If we find a better copy in the future, we will add it to the Agilent website.

Support for Your Product

Agilent no longer sells or supports this product. You will find any other available product information on the Agilent website:

www.agilent.com

Search for the model number of this product, and the resulting product page will guide you to any available information. Our service centers may be able to perform calibration if no repair parts are needed, but no other support from Agilent is available.

Agilent Technologies

HEWLETT PACKARD

Application Note 339-1

Impedance Characterization of Resonators Using the HP 4194A Impedance/Gain-Phase Analyzer



1. Resonator Evaluation

A wide variety of resonators are now used in an ever increasing spectrum of products such as microprocessor clock oscillators, tone generators TV IF filters ... etc. The types of resonators used include crystal, ceramic, polymer, mechanical and ferrite.

Until the HP 4194A became available, serious analysis and testing of resonators, and determining the device parameters from measured frequency characteristics required an external computer/controller to perform complex calculations and control functions. The HP 4194A is a single instrument solution for obtaining frequency characteristics and computing parameters. Frequency characteristics are displayed on a color CRT and markers are used to define an area to be analyzed and extract data from the point indicated by the marker. Data analysis is carried out using the HP 4194A's computational, programmability and equivalent circuit analysis functions We will discuss an efficient method to evaluate ceramic resonators as the example. The single instrument solution to complex problems!

2. Resonator Characterization Using the HP 4194A

(1) Resonant/Antiresonant Frequency

The resonant/antiresonant frequencies (series and parallel resonance modes) are the principle parameters of interest when analyzing resonators. The 4194A's markers are used to zoom in on an area of the displayed frequency characteristics to quickly find the points of resonance.

Markers are used to read frequency, impedance and phase anywhere on the displayed trace. Measurement resolution of 1mHz enables you to easily detect abrupt changes in frequency characteristics such as found in crystal resonators.



Figure 1-1



Figure 1-2

(2) Electro Mechanical Coupling Coefficient

This parameter indicates the efficiency of electrical to mechanical energy conversion. The coupling coefficient is calculated from the resonant/ antiresonant frequency data using the following equation.

$$Kt = \left(\frac{\pi}{2} \cdot \frac{fr}{fa} \tan\left(\frac{\pi}{2} \cdot \frac{fa - fr}{fa}\right)\right)^{1/2}$$

where kt is the electro-mechanical coupling coefficient, fr is the resonant frequency, and fa is the antiresonant frequency.

Arithmetic operations such as used in the above equation are possible from the HP 4194A's front panel, and the marker function can be used to specify the values for fr and fa from the displayed measurement data.



Figure 2-1

(3) Equivalent Circuit Analysis

The equivalent circuit model for a resonator is shown in Figure 3-1. The components used in this model, L, Ca, R and Cb, are the basic elements needed to accurately model a resonator over the 4194A's frequency range. The 4194A computes the values of the equivalent circuit components from the measured data. The equivalent circuit analysis function is a powerful tool, unique to the 4194A, that designers can use to vary circuit constants when simulating possible changes in design, processing tolerances, and temperature. In a matter of minutes an engineer can measure the responce of a resonator, compute the values of the equivalent circuit components, and display the response of the hypothetical resonator simultaneously with the resonator's measured response (Figure 3-2).



Figure 3-1



Figure 3-2

(4) Circle Diagram of Admittance

The circle diagram of admittance is a quick and convenient method of evaluating resonators. The diameter of the admittance circle represents the Q of the resonator, and the closer the admittance circle comes to forming a perfect circle, the better the stability of the resonator.

Previously, obtaining an admittane circle diagram required the use of an X-Y recorder or an external computer. You can obtain an admittance circle diagram directly without other instruments or a computer. You can use the 4194A's marker function to read the resonant frequency or other information from the admittance diagram.



Figure 4-1

3. Automatic Evaluation of Resonators

The HP 4194A's internal programming function, Auto Sequence Program (ASP), gives you the ability to perform automatic evaluation without the need of an external computer.

ASP can control all of the HP 4194A's operations: measurement, display, and analysis. ASP can automate any of the foregoing resonator evaluations, for quick and efficient evaluation using only a single instrument. Figure 5 shows a sample program to perform quick automated resonator evaluation by measuring the following items.

(1) Resonant/antiresonant frequency
(2) Electro Mechanical coupling coefficient
(3) Equivalent circuit analysis
(4) Circle diagram of admittance

```
 10 RST
 20 SWM2
 30 NOP=401
 40 CMT"AN339-1;CERAMIC RESONATOR EVALUATION"
 50 DISP "CONNECT 16047D"
 60 BEEP
 70 PAUSE
 80 DISP "CONNECT DEVICE"
 90 BEEP
100 PAUSE
110 DISP "SEARCHING APPROPRIATE RANGE"
120 CENTER=30 MHZ
130 SPAN=1 MHZ
140 SWTRG
150 ASC2;AUTOA
160 MKMNA
170 START=MKR-50E3
180 SWTRG
190 MKMXA
200 STOP=MKR+50E3
210 SWTRG
220 CMT"FREQUENCY CHAR. OF CERAMIC RESONATOR"
230 MKR=CENTER
240 DISP "PRESS <CONT> SOFTKEY"
250 BEEP
260 PAUSE
270 MKMNA;R1=MKR
280 CMT"RESONANT FREQUENCY"
290 DISP "PRESS <CONT> SOFTKEY"
300 BEEP
310 PAUSE
320 MKMXA;R2=MKR
330 CMT"ANTIRESONANT FREQUENCY"
340 DISP "PRESS <CONT> SOFTKEY"
350 BEEP
360 PAUSE
370 MCF5;MKR=R2
380 SMKR=R1
390 CMT"ELECTRO MECHANICAL COUPLING COEFFICIENT"
400 R3=SQR(PI/2*R1/R2*TAN(PI/2*(R2-R1)/R2))
410 DISP R3
420 BEEP
430 PAUSE
440 CMT"EQUIVALENT CIRCUIT ANALYSIS"
450 EQDSP
460 EQC5
470 EQCAL
480 DISP "PRESS <CONT> SOFTKEY"
490 BEEP
500 PAUSE
510 CMT"FREQUENCY CHARACTERISTICS SIMULATION"
520 FCHRS
530 DISP "PRESS <CONT> SOFTKEY"
540 BEEP
550 PAUSE
560 MKMNA
570 CENTER=MKR
580 IMP9;DSP2;DPAB1;ASC1
590 SWTRG
600 AUTO
610 MKR=R2
620 CMT"ADMITTANCE CIRCLE"
630 BEEP
640 PAUSE
650 CMT"AN339-1:CERAMIC RESONATOR EVALUATION"
660 DISP "END"
670 BEEP
680 END
```

Figure 5

4. Ordering Information

All of the functions and evaluations we have described can be performed using the HP 4194A Impedance/Gain-Phase Analyzer and the HP 16047D Test Fixture furnished with the 4194A.

When you order the HP 4194A, you must indicate which test port impedance option you want. Option 350 is for a test port impedance of 50 ohm, and Option 375 is for a test port impedance of 75 ohm. Other test fixtures are availabe. Contact the nearest HP Sales Office for details.

HEWLETT PACKARD

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, Rue du Bois-du-Lan, P.O. Box, CH-1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

Printed in Japan October, 1985 5950-2882